MODELING OF OLIGOPOLISTIC PRODUCER BEHAVIOR WITH GAMS

ERWIN KALVELAGEN

Abstract. This document describes a simple complementarity model formulated in GAMS.

1. Oligopolistic producer behavior

An oligopoly is a situation where there are a limited number of suppliers. As a result the behavior of a single supplier has impact on the price. One could say that an oligopoly lies somewhere between the extremes formed by a monopolistic environment and the case of perfect competition [5].

Assume there are m firms, each with the following production cost function [3, 1]:

$$f_i(q_i) = c_i q_i + \frac{\beta_i}{\beta_i + 1} K_i^{-1/\beta_i} q_i^{(\beta_i+1)/\beta_i} \tag{1}$$

where q_i is the production volume[1]. Furthermore assume a demand price function or inverse demand function of the form:

$$\begin{aligned} p(Q) &= 5000^{1/1.1} Q^{-1/1.1} = A Q^{-1/1.1} \\ Q &= \sum_{i=1}^{m} q_i \end{aligned} \tag{2}$$

I.e. the price depends on the total production.

Each producer will try to maximize it's profit defined by

$$u(q_i) = p(Q) q_i - f_i(q_i) \tag{3}$$

The optimality conditions can be stated as follows:

$$u'(q_i) \le 0 \perp q_i \ge 0 \tag{4}$$

where

$$\begin{aligned} u'(q_i) &= \frac{\partial u(q_i)}{\partial q_i} \\ &= p(Q) + q_i p_i'(Q) - f_i'(q_i) \\ &= p + q_i \left[-\frac{A}{1.1} Q^{-2.1/1.1} \right] - c_i - \left(\frac{q_i}{K_i} \right)^{1/\beta_i} \end{aligned} \tag{5}$$

Date: March 12, 2003, Revised May 22, 2009.

[1] [1] contains a typographical error in the specification of this function



FIGURE 1. Production cost function for firm $i = 1$

To simplify the model we use auxiliary equations to calculate $Q = \sum_i q_i$ and $p = p(Q)$, resulting in the complete model:

$$
\begin{aligned}
p + q_i \left[-\frac{A}{1.1} Q^{-2.1/1.1} \right] - c_i - \left(\frac{q_i}{K_i} \right)^{1/\beta_i} &\leq 0 \perp q_i \geq 0 \\
Q = \sum_i q_i &\perp Q \text{ free} \\
p = p(Q) &\perp p \text{ free}
\end{aligned} \tag{6}
$$

This is a non-linear complementarity model, which can be conveniently solved in GAMS. Several assumptions need to be made, listed in [2]:

- $p(Q)$ is non-negative and decreasing for $Q \geq 0$
- $Qp(Q)$ is a proper concave function for $Q \geq 0$
- f_i are positive, convex and increasing
- There is a $\bar{q}_i \geq 0$ such that $\bar{q}_i p(\bar{q}_i) - f_i(\bar{q}_i) < 0$

2. MODEL OLIGOPOLY.GMS

[2]

This model is similar to the model `oligomcp.gms` in the GAMS model library.

```
$ontext

  Oligopolistic producer behavior

```

[2] http://www.amsterdamoptimization.com/models/micro/oligopoly.gms



FIGURE 2. Inverse demand function $p(Q) = 5000^{1/1.1}Q^{-1/1.1}$

```
  Erwin Kalvelagen, december 2001.

  References:

      Murphy F H, H.D. Sherali and A.L.Soyster "A mathematical
      programming approach for determining oligopolistic market
      equilibrium" Mathematical Programming 24 (1982) pp. 92-106.

      Harker P T, "Oligopolistic equilibrium", Mathematical
      Programming 30 (1984) pp. 105-111.

$offtext

set i 'firms' /firm1*firm5/;

table data(i,*) 'production cost function paramaters'
           c    K    beta
  firm1   10    5    1.2
  firm2    8    5    1.1
  firm3    6    5    1.0
  firm4    4    5    0.9
  firm5    2    5    0.8
;
parameter c(i),K(i),beta(i);
c(i) = data(i,'c');
K(i) = data(i,'K');
beta(i) = data(i,'beta');

scalar A 'constant in demand function';
A = 5000**(1/1.1);

variables
   q(i)       'quantities produced'
   p          'demand price'
   tq         'total q'
```

```
;
positive variables q;

equations
   equilibrium(i)
   demand_function  'auxiliary equation'
   summation        'auxiliary equation'
;

* production cost function:
* f(i) =e=
*    c(i)*q(i) + [beta(i)/(beta(i)+1)]*[K(i)**(-1/beta(i))]*[q(i)**((beta(i)+1)/beta(i))];


summation..
    tq =e= sum(i,q(i));

demand_function..
    p =e= A*[tq**(-1/1.1)];

equilibrium(i)..
    0 =g= p + q(i)*[A*(-1/1.1)*tq**(-2.1/1.1)]
          -c(i)-(q(i)/K(i))**(1/beta(i));



*
* initial values
*
q.l(i) = 10;
tq.l = sum(i,q.l(i));
p.l = A*[tq.l**(-1/1.1)];

option mcp=path;
model oligopoly /demand_function.p, equilibrium.q, summation.tq/;
solve oligopoly using mcp;
```

This problem can also be conveniently formulated as a Variational Inequality model [2, 4]: find $x \geq 0$ such that

$$\begin{aligned} &\langle F(x), z - x\rangle \geq 0 \text{ for all } z \geq 0 \\ &F_i(q) = -\left[p(Q) + q_i p'(Q) - c_i'(q_i)\right] \end{aligned} \tag{7}$$

Formulating this model as an optimization model is not trivial [3]. The model that maximizes $\sum_i u(q_i)$ is certainly not appropriate. It would describe the monopolistic situation when the firms are merged. We can compare the results for this situation and the model above as follows.

3. Model oligopoly2.gms

[3]

```
$ontext

  Oligopolistic producer behavior

  Erwin Kalvelagen, december 2001.

  References:

      Murphy F H, H.D. Sherali and A.L.Soyster "A mathematical
      programming approach for determining oligopolistic market
      equilibrium" Mathematical Programming 24 (1982) pp. 92-106.

      Harker P T, "Oligopolistic equilibrium", Mathematical
      Programming 30 (1984) pp. 105-111.
```

[3] http://www.amsterdamoptimization.com/models/micro/oligopoly2.gms

```
$offtext

set i 'firms' /firm1*firm5/;

table data(i,*) 'production cost function paramaters'
          c    K    beta
  firm1   10    5    1.2
  firm2    8    5    1.1
  firm3    6    5    1.0
  firm4    4    5    0.9
  firm5    2    5    0.8
;
parameter c(i),K(i),beta(i);
c(i) = data(i,'c');
K(i) = data(i,'K');
beta(i) = data(i,'beta');

scalar A 'constant in demand function';
A = 5000**(1/1.1);

variables
   q(i)       'quantities produced'
   p          'demand price'
   tq         'total q'
;
positive variables q;

equations
   equilibrium(i)
   demand_function  'auxiliary equation'
   summation        'auxiliary equation'
;

* production cost function:
* f(i) =e=
*    c(i)*q(i) + [beta(i)/(beta(i)+1)]*[K(i)**(-1/beta(i))]*[q(i)**((beta(i)+1)/beta(i))];


summation..
    tq =e= sum(i,q(i));

demand_function..
    p =e= A*[tq**(-1/1.1)];

equilibrium(i)..
    0 =g= p + q(i)*[A*(-1/1.1)*tq**(-2.1/1.1)]
          -c(i)-(q(i)/K(i))**(1/beta(i));


*
* initial values
*
q.l(i) = 10;
tq.l = sum(i,q.l(i));
p.l = A*[tq.l**(-1/1.1)];

option mcp=path;
model oligopoly /demand_function.p, equilibrium.q, summation.tq/;
solve oligopoly using mcp;


*------------------------------------------------------------------
* now formulate the situation where the firms are merged and a
* single large monopoly is formed.
*------------------------------------------------------------------

equations
   objective            'monopolistic objective'
   profit_function(i)   'calculate profit for each firm'
   cost_function(i)     'calculate cost for each firm'
;
```

```
variables
   obj                 'objective variable'
   u(i)                'utility: profit'
   f(i)                'cost'
;

objective.. obj =e= sum(i,u(i));

profit_function(i).. u(i) =e= p*q(i) - f(i);

cost_function(i).. f(i) =e=  c(i)*q(i) +
       [beta(i)/(beta(i)+1)]*[K(i)**(-1/beta(i))]*[q(i)**((beta(i)+1)/beta(i))];

model monopoly /objective,profit_function,cost_function,summation,demand_function/;
solve monopoly maximizing obj using nlp;
```

	oligopoly	monopoly
p	18.3	80.3
q_1	36.9	0
q_2	41.8	0
q_3	43.7	6.5
q_4	42.7	14.7
q_5	39.2	19.0
$\sum q_i$	204.3	40.1

TABLE 1. Results comparing oligopoly and monopoly

The results comparing the oligopolistic situation with a profit maximizing monopolist are shown in table 1. Both the prices and quantities show the expected patterns reflecting how the firms behave.

The natural formulation of this problem as an MCP model illustrates the popularity of MCP solvers under economists.

REFERENCES

1. Patrick T. Harker, *A variational inequality approach for the determination of oligopolistic market equilibrium*, Mathematical Programming **30** (1984), 105–111.
2. Patrice Marcotte, *Advantages and drawbacks of variational inequality formulations*, Tech. report, Département d'informatique et de recherche opérationelle, Université de Montréal, Montréal, Canada, July 1994.
3. Frederic H. Murphy, Hanif D. Sherali, and Allen L. Soyster, *A mathematical programming approach for determining oligopolistic market equilibrium*, Mathematical Programming **24** (1982), 92–106.
4. Anna Nagurney, *Network economics, a variational inequality approach*, revised second ed., Advances in Computational Economics, vol. 10, Kluwer Academic Publishers, 1999.
5. Hal R. Varian, *Intermediate microeconomics. a modern approach*, fifth ed., W. W. Norton & Co., 1999.

AMSTERDAM MODELING GROUP, WASHINGTON DC/THE HAGUE
E-mail address: erwin@amsterdamoptimization.com